
SA305 – Linear Programming Spring 2013
Asst. Prof. Nelson Uhan

Lesson 11. Resource Allocation Models, Revisited

1 Writing optimization models with symbolic input parameters

Problem 1. Farmer Jones decides to supplement his income by baking and selling two types of cakes, chocolate
and vanilla. Each chocolate cake sold gives a pro�t of $3, and the pro�t on each vanilla cake sold is $5. Each
chocolate cake requires 20 minutes of baking time and uses 4 eggs and 4 pounds of �our, while each vanilla
cake requires 40 minutes of baking time and uses 2 eggs and 5 pounds of �our. Farmer Jones has 260 minutes
of baking time, 32 eggs, and 40 pounds of �our available. Assume all cakes baked are sold, and fractional cakes
are OK. Write a linear program that determines how many of each type of cake should Farmer Jones bake in
order to maximize his pro�t.

Recall that the linear program we wrote for this problem is

C = number of chocolate cakes to bake
V = number of vanilla cakes to bake

maximize 3C + 5V (total pro�t)
subject to 20C + 40V ≤ 260 (baking time available)

4C + 2V ≤ 32 (eggs available)
4C + 5V ≤ 40 (�our available)
C ≥ 0,V ≥ 0

Problem 2. Farmer Jones decides to supplement his income by baking and selling cakes. Let K be the set of
cake types that he sells. Each cake k sold yields a pro�t of pk , for all k ∈ K. Each cake type requires a certain
mixture of ingredients. Let I be the set of ingredients that are used. Each type k cake requires aik units of
ingredient i, for all i ∈ I and k ∈ K. Farmer Jones has bi units of ingredient i available, for all i ∈ I. Assume all
cakes baked are sold, and fractional cakes are OK. Write a linear program that determines how many of each
type of cake should Farmer Jones bake in order to maximize his pro�t.

● Recall that input parameters are quantities that are given and �xed

● What are the input parameters in Problem 2?

● How do these input parameters relate to the those given in Problem 1?

● Write a linear program for Problem 2, using summation notation and for statements.

● �is model has symbolic input parameters

○ “Placeholders” for actual set elements and numerical values

● �is model is valid for any problem of the same structure

○ Just need to specify actual set elements and numerical values for the symbolic input parameters
○ e.g. Specify elements for K and I; numerical values for pk for k ∈ K, bi for i ∈ I,
and aik for i ∈ I and k ∈ K

2 Sets, summations, for statements, and symbolic input parameters inMathProg

● How do we use sets, summation notation, for statements, and symbolic input parameters inMathProg?

maximize ∑k∈K pkxk
subject to ∑k∈K a i ,kxk ≤ b i for i ∈ I

xk ≥ 0 for k ∈ K

Model �le with symbolic input parameters
Chocolate Vanilla Total avail.

Baking Time 20 40 260
Eggs 4 2 32
Flour 4 5 40
Pro�t 3 5

Data �le with set elements
and numeric values for input parameters

GLPK/GUSEK

maximize 3C + 5V
subject to 20C + 40V ≤ 260

4C + 2V ≤ 32
4C + 5V ≤ 40
C ≥ 0,V ≥ 0

Model with set elements and numeric values
for input parameters, solved by GLPK

2



● MathProgmodel �le (farmerjones.mod)

## Input parameters ##

set K; # set of cake types

set I; # set of ingredients

param p{k in K}; # p[k] = profit for cake type k

param b{i in I}; # b[i] = amount of ingredient i available

param a{i in I, k in K}; # a[i,k] = amount of ingredient i used in 1 type k cake

## Decision variables and variable bounds ##

var x{k in K} >= 0; # x[k] = number of type k cakes to produce

## Objective function ##

# Maximize total profit

maximize total_profit:

sum{k in K} p[k] * x[k];

## General constraints ##

# Amount of ingredient i used <= amount of ingredient i available

subject to ingredient_avail{i in I}:

sum{k in K} a[i,k] * x[k] <= b[i];

end;

● MathProg data �le for the Problem 1 (farmerjones-original.dat)

# Data for the original Farmer Jones problem

# Set of cake types

set K := Chocolate Vanilla;

# Set of ingredients

set I := BakingTime Eggs Flour;

# p[k] = profit for cake type k

param p :=

Chocolate 3

Vanilla 5;

# b[i] = amount of ingredient i available

param b :=

BakingTime 260

Eggs 32

Flour 40;

# a[i,k] = amount of ingredient i used in 1 type k cake

# rows correspond to i, columns correspond to k

param a:

Chocolate Vanilla :=

BakingTime 20 40

Eggs 4 2

Flour 4 5;

end;

● Running the model and data �le in combination in GUSEK:

○ Make sure farmerjones.mod is the only model �le open
○ Swtich to farmerjones-original.dat
○ Select Tools→ Set as Default .dat File

○ Switch to farmerjones.mod
○ Make sure Tools→ Generate Output File on Go is checked
○ Select Tools→ Go

○ You can check if the model and data combine in the way you expect by selecting
Tools→ Build Cplex LP

◇ Note that in a Cplex LP �le, variables are assumed to be nonnegative unless otherwise speci�ed
◇ Do not follow this practice! Always specify nonnegativity constraints if necessary!

Problem 3. Farmer Jones’s cake business has been quite successful! With some new recipes in hand, he is
trying to determine how to expand his cake o�erings. Farmer Jones can now bake and sell 3 types of cakes:
chocolate, vanilla, red velvet. Each cake requires varying amounts of 4 ingredients: prep time, baking time,
eggs, �our. In particular, the amount of each ingredient needed in each type of cake is given below:

Chocolate Vanilla Red Velvet
prep time 30 20 50
baking time 25 40 35

eggs 3 2 4
�our 4 4 5

Each chocolate cake generates a pro�t of $4, vanilla $5, and red velvet $5. Farmer Jones has 240 minutes
of prep time, 280 minutes of baking time, 50 eggs, and 40 pounds of �our available. Write a data �le
farmerjones-new.dat that accompanies the model �le farmerjones.mod to solve Farmer Jones’s new
problem. Solve the linear program.

